
Chapter 30.  Xbasic control of Form Properties 
Imagine that you have a customer form, and you want to examine all of that customer's invoices.  You start by 
creating a Customer history form based on a one:many set from customer to invoice_header. You then want 
to place a button on your standard Customer information form to open the Customer history form 
synchronized to the current customer.  Alpha Five's button genie will help you create a button whose OnPush 
script is: 

'Check to see if you are running the script from within a Form
dim flag_error as l
flag_error = .f.
if is_object(topparent.this) then 

p = topparent.this
if p.class() = "Form" then 

DELETE Parameter1 
'Set the Parameter variable to the object's .value property
Parameter1 = eval("P:Customer_id.value")

else    'Not in a Form. This is an error condition.
flag_error = .t.

end if 
else

flag_error = .t.
end if 

if flag_error then 
ui_msg_box("Error","You can only run this script from within a Form",UI_STOP_SYMBOL)
end 

end if 

filter = "Customer_id = [varC->parameter1]"
query.filter = replace_parameters(filter,local_variables())
query.order = ""

'Open the layout_type showing just the records that satisfy the filter
DIM Shared varP_Customer_history as P
DIM layout_name as c 
layout_name = "Customer history"

dim tempP as p
'Get pointer to existing window. In case layout_name is qualified with a dictionary name, extract up 
to first @. In case formname has spaces, normalize it
tempP=obj(":"+object_name_normalize(word(layout_name,1,"@")))
'Test if pointer is valid

if is_object(tempP) then 
'Test if pointer refers to a form or browse
if tempP.class() = "form" .or. tempP.class() = "browse" then 

'Apply the query to the form or browse as the Base query. (this means that 
'the user cannot turn the query off, only add to the query)
tempP.BaseQueryRun(query.filter,query.order)
'then activate the already open window
tempP.activate()

else
'Window is not already open, so open it
varP_Customer_history = :Form.viewqueried(layout_name,query.filter, query.order )

end if
else 

varP_Customer_history = :Form.viewqueried(layout_name,query.filter, query.order )

end if
This is a pretty ugly script, but don't despair. Almost half of it is just error checking to be sure that you haven't 
opened the script in some silly manner, such as from the Control Panel. Let's eliminate some of the excess 
baggage and see what it reduces to:

p = topparent.this

query.filter = "Customer_id="+quote(P:Customer_id.value)
query.order = ""

DIM f  as P
DIM layout_name as c 
layout_name = ":Customer_history"

f = :Form.viewqueried(layout_name,query.filter, query.order )

This script does some extremely powerful manipulation.  Think of this Xbasic script as an invisible user C call 
her Sue. In line 10, Sue clicks on the control-panel object form and invokes its viewqueried method to get the 
Customer history form from the current database.  In lines 3 and 4, Sue has pre-entered a filter expression for 
the form, which is used by viewqueried.

Placing a query on a child table 

I mentioned that the button genie will create this script for you.  But this book aims to take you farther than the 
button genies can go!  The button genie will let you create a single filter expression for a single table.  What if 
you want to go further and restrict invoices to those created in 1996?  We can use Xbasic to create filters for 
more than one table.  We can get a pointer to the Customer history form's invoice_header table with the 
syntax
f:Tables:invoice_header
We can then set a query for the child table (invoice_header) using the following code:

p=topparent.this

query.filter = "Customer_id="+quote(P:Customer_id.value)
query.order = ""

DIM f  as P
DIM layout_name as c 
layout_name = ":Customer_history"

f = :Form.viewqueried(layout_name,query.filter, query.order )
filter="between(date,{1/1/1996},{12/31/1996})"
order="date"
f:Tables:invoice_header.filter_expression=filter
f:Tables:invoice_header.order_expression=order
f:Tables:invoice_header.query()
f.show()
f.activate()

The bolded lines are the new ones. 

Preventing any changes to a form 
Perhaps you want to prevent the user from making any changes on your form.  That is simple in Xbasic.  The 
new properties we will use are:
<form_pointer>.restrict_enter
<form_pointer>.restrict_change
which are akin to changing the properties of the form in design mode, and the property
browse_name.browse.readonly
which changes the editing properties of a browse.
Here is the script:
p = topparent.this

filter = "Customer_id = "+quote(Customer_id.value)
query.filter = filter
query.order = ""
'Open the layout_type showing just the records that satisfy the filter
DIM Shared f as P
DIM layout_name as c 
layout_name = "Customer history"

f = :Form.viewqueried(layout_name,query.filter, query.order,"hidden" )
f:Tables:invoice_header.filter_expression="between(date,{1/1/96},{12/31/96})"
f:Tables:invoice_header.order_expression="date"
f:Tables:invoice_header.query()

f.restrict_enter=.t. ' No enters allowed at form level
f.restrict_change=.t. ' No changes allowed at form level
f:browse1.browse.readonly=.t.' No enters or changes allowed for invoice header browse
f.show()

